Dopo le routine, tocca alle funzioni: la differenza tra una routine e una funzione è che la prima esegue un certo numero di istruzioni, la seconda esegue un certo numero di istruzioni e in più restituisce un valore, che quindi può essere memorizzato in una variabile. L’uso delle funzioni è comodo quando è opportuno sapere quale sia il risultato finale delle operazioni eseguite: ad es., la funzione potrebbe restituire un valore di errore se qualcosa è andato storto durante la sua esecuzione, oppure potrebbe restituire il risultato di un’operazione matematica come quella eseguita dalla routine quadrato che abbiamo visto nella lez. precedente. Per dichiarare una funzione bisogna usare la parola chiave Function al posto di Sub; inoltre è sempre meglio, se possibile, specificare il tipo di dati restituito dalla funzione. Ad es.:

private function quadrato (param as integer) as long

quadrato=param^2

debug.print quadrato

end function

Il valore che la funzione deve restituire deve essere assegnato al nome della funzione (quadrato=param^2), che viene quindi trattato come se fosse una variabile: perciò rispetto alla routine quadrato (v.lez. precedente) si può fare a meno di dichiarare una variabile locale a cui assegnare il risultato dell’elevamento a potenza. Inoltre, possiamo fare a meno di usare l’istruzione debug.print all’interno della funzione, perché il valore che vogliamo visualizzare è quello restituito dalla funzione e quindi è visibile anche all’esterno di essa: nell’evento load del form, anziché scrivere

quadrato 2

quadrato 5

quadrato 8

possiamo quindi scrivere:

debug.print quadrato(2)

debug.print quadrato(5)

debug.print quadrato(8)

In questo caso è necessario usare le parentesi tonde: queste vanno sempre indicate se si vuole che la funzione restituisca effettivamente un valore; infatti è anche possibile richiamare la funzione senza parentesi:

quadrato 2

come se fosse una routine, ma non sarà possibile scrivere:

debug.print quadrato 2

perché chiamando la funzione senza usare le parentesi si impedisce che essa restituisca un valore. Lo stesso accade se si chiama la funzione con l’istruzione call; pertanto mentre una routine può essere chiamata indifferentemente con o senza le parentesi, una funzione deve essere chiamata con le parentesi (ma senza usare l’istruzione call) affinché restituisca un valore.

Per quanto riguarda il passaggio di parametri, valgono le stesse regole che abbiamo visto per le routine; a questo proposito c’è da sapere un’altra cosa: i parametri possono essere preceduti dalla parola chiave byval o byref. Queste parole chiave specificano, rispettivamente, che il parametro viene passato “per valore” o “per riferimento”: senza entrare troppo nei dettagli, è sufficiente dire che nel primo caso la funzione (o la routine) conosce soltanto il valore della variabile passata come argomento, mentre nel secondo caso conosce, per così dire, la variabile stessa, potendo quindi intervenire direttamente su di essa. Concretamente ciò significa che se la variabile è passata per valore, la funzione (o la routine) potrà eseguire operazioni sul valore della variabile ma non potrà modificarla; se invece la variabile è passata per riferimento, la funzione potrà modificare direttamente la variabile. Un esempio chiarirà tutto: creiamo una funzione con due argomenti, uno passato per valore, l’altro per riferimento e facciamo qualche operazione su di essi:

private function prova (byval valore as integer, byref riferimento as integer) as integer

valore=valore+1

riferimento=riferimento+1

prova=valore+riferimento

end function

La funzione non fa altro che incrementare di 1 i due argomenti e restituirne la somma. Ora dichiariamo, nell’evento load del form, due variabili e passiamole alla funzione:

dim var1 as integer, var2 as integer

dim risultato as integer

var1=3

var2=10

risultato=prova(var1,var2)

debug.print risultato

debug.print var1, var2 

Eseguendo il progetto, noteremo che nella finestra immediata compaiono i valori:

15

3 11

15 è il risultato della funzione ((3+1)+(10+1)=15); la variabile var1, che è stata passata per valore e che quindi non ha potuto essere stata modificata, conserva il valore che aveva prima della chiamata della funzione, cioè 3; invece var2, che è stata passata per riferimento, è stata effettivamente modificata durante l’esecuzione della funzione, e infatti dopo la chiamata ha assunto il valore 11, cioè il valore originario più 1. Se non viene specificata né byval né byref, Visual Basic penserà automaticamente che la variabile è stata passata per riferimento: questa infatti è l’impostazione predefinita.

Un’altra parola chiave che può precedere gli argomenti è Optional: questa specifica che l’argomento seguente non è obbligatorio, e in tal caso è possibile indicare un valore di default che l’argomento deve assumere nel caso in cui l’istruzione chiamante la funzione (o la routine) ometta effettivamente il parametro. Ad es.:

private function prova (a as integer, optional b as integer=1) as integer

prova=a+b

end function

se nell’evento load del form scriviamo:

debug.print prova(2)

debug.print prova(2,3)

vedremo nella finestra immediata:

3

5

Nel primo caso, infatti, il parametro b è omesso e quindi assumerà per default il valore 1.

Come per le routine, anche le variabili locali di una funzione possono essere dichiarate “static” per indicare che il loro valore viene mantenuto tra una chiamata e l’altra della funzione (v.lez. 2): se si vuole che TUTTE le variabili locali siano statiche, si può dichiarare la funzione in questo modo:

private static function prova() as integer

dim a as string ‘variabile statica

dim b as long ‘variabile statica

end function

Le variabili locali saranno statiche anche se nella loro dichiarazione non è stata specificata la parola chiave static.

